


SPECTRAVIDEO™

AUSTRALASIAN USERS GROUP



News Letter

REGISTERED BY AUSTRALIA POST PUBLICATION NO. TBH 0917 CATEGORY "B"

CONTENTS	PAGE
EDITORIAL COMMENT	2
A.P.C. 84. (S-V)	3
EXTRA COLORS (program)	4
PAINT (program)	4
EXPLORING BASIC Pt-6	5
LISTING 1	8
LISTING 2	9
LISTING 3	9
COPY PATCH	10
PROGRAM OF THE MONTH (cycle)	11
S-V RETAILERS ??	13
GRAPHICS TABLET	14
LIBRARY NOTES	15

ISSUE No.
1 - 12
DATE
SEPTEMBER-84

ALL CORRESPONDENCE TO:

S. A. U. G.,
P. O. BOX 191,
SOUTH LAUNCESTON,
TASMANIA, 7249.

(083) 312648

MEMBERSHIP FEES

AUSTRALIA \$15.00
OVERSEAS \$20.00
OVERSEAS AIRMAIL .. \$25.50

SPECTRAVIDEO™

AUSTRALASIAN USERS GROUP

EDITORIAL COMMENT

Some major steps forward have occurred this month with the Newsletter that are worth mentioning.

Firstly the newsletter is now registered by Australia Post. What does this mean? Two things; One postage is cheaper, and Two because of this we DO NOT need to increase membership fees in the coming 12 Months (We were seriously considering increasing the membership). I hope you all appreciate the value for money you are receiving, at only \$15 per year the Newsletter is the CHEAPEST S.V. newsletter in Australia and in comparison to other Computer Groups we are also value for money.

Next in an ever increasing attempt to improve the quality of the Newsletter we are now able to list all programs with **GRAPHICS** directly to the printers and so no more hand drawn graphics. Last month is a good example of what I mean.

EXAMPLE:

```
1270 LOCATE1+Q*8,16:PRINT" \-----\ ":FORL=17TO21:LOCATE1+Q*8,L:PRINT" |      |":N
      EXTL
1280 LOCATE1+Q*8,22:PRINT" \-----/ ":RETURN
1310 PRINT"#####";GOTO1330
1410 LOCATE2+Q*8,17:PRINTUSING"\ \ ♣";A$(A(Q)):LOCATE2+Q*8,21:IFA(Q)<>10THEN
      PRINTUSING"♣  !";A$(A(Q))ELSEPRINT"♣ 10"
```

Note: that the new listings are different to those before in more than one way. Apart from the **GRAPHICS** we also now *indent* lines that are too long (longer than 80 characters) so that reading and entering of the program into your computer is easier. So remember when a line is indented in our listings **DON'T** add the spaces when you are typing just continue on from under the beginning of the previous line.

Modems

I am now the proud owner of a DATAPHONE II modem and it works just great. (See The Australian Beginning) Also MODEM software is now available from the library, so I am looking forward to hearing from any other member with a Modem, (or planning to buy one).

CPM 2.22

One more thing I have found out, there are two CP/Ms in Australia for the S.V. You can tell which one you have by checking the sign-on message. It will say CP/M 2.2 or CP/M 2.22. 2.22 is the newest and has many bugs removed from 2.2. (well worth it if you can get your hands on it). It mainly supports the RS-232, and PRINTER better, but has other improvements. Any CP/M patches from now on will support both versions.

Many thanks to all who contributed to this Newsletter. Next month is our Birthday (1 Year old) so look forward to a bumper issue with extra programs.

SPECTRAVIDEO™

AUSTRALASIAN USERS GROUP

A.P.C. 84. (S-V)

By M.C.Dodd.

We recieved many letters from members that attended the A.P.C. Computer Show. Due to space restrictions we can't print them all so here is just one.


I arrived at the APC 84 Computer Show early afternoon on the first day and was surprised to see that the Spectravideo stand was still being set up.....about half of the computers on the stand being run into black and white monitors and those running color monitors having problems maintaining a good display.....GREMLINS perhaps. Some very puzzled spectators and myself wondering what was going on.

The new Super Expander was on display, running CP/M with an 80 column card into a B/W monitor. The unit itself looked very good and appeared to be good quality. The one I saw had an inbuilt Centronics Interface, twin disk controller card and just one disk drive. The disk drive appeared to be just as noisy as previous Spectravideo units.

On handout at the stand were the advertising leaflets on the new Mk II computers. Look good on paper and are made with current technology, have fewer 'chips', an inbuilt modulator, and are supposed to be cheaper than the old models.

Overall, I was not impressed with the Spectravideo Stand or the presentation.....I stress that this is a personal opinion only and in no way should be taken to represent the views of any one or any group other than myself.

To the 'RUMOURS' department now.....It would appear that IMAGINEERING, EPYX, and BRODERBUND are all about to release software for the SV computers.....if true it could be the boost needed. Also not confirmed but strong is the news of the MSX adaptor to convert our MSX machine to MSX standard, for not only is the Spectravideo the first MSX machine to be released, it appears that it is the first MSX machine to need an adaptor to make it MSX compatible. Yes folks, read the paragraph again, it means exactly what it says !.



SPECTRAVIDEO™

AUSTRALASIAN USERS GROUP

EXTRA COLORS

by : W SZAPIRKO

```
10 '
20 ' EXTRA COLOURS FOR EV
30 '
40 ' BY W.SZAPIRKO
50 '
60 ' WORKS ONLY WITH COLOUR VDU
70 '
80 '
100 SCREEN 1
110 FOR T=0 TO 246 STEP 16
120 C=(T\16)
130 LINE(T,0)-(T+15,192),C,BF
140 NEXT T
150 FOR T=0 TO 191 STEP 2
160 C=(T/12)
170 LINE(0,T)-(256,T),C
180 NEXT T
190 GOTO 190
```

PAINT

by : ?????

```
100 SCREEN 1
110 PSET (128,96)
120 G=0
130 FOR T=0 TO 148 STEP 1.57
140 X=COS(T)*G+128
150 Y=SIN(T)*G+96
160 G=G+1
170 LINE-(X,Y),2
180 NEXT T
190 DRAW "R3"
200 PAINT (128,97),2
210 FOR G=1 TO 1000:NEXT G
220 GOTO 220
```

These two programs "EXTRA COLORS & PAINT" may give you a bit of fun during a quite moment.

SPECTRAVIDEO™

AUSTRALASIAN USERS GROUP

EXPLORING BASIC Pt-6

By L.A. Dunning.


This month, I describe SCREEN 2 and how SPRITES are used.

MULTICOLOR (SCREEN 2)


SCREEN 2 works in an entirely different manner than that of SCREEN 1. When I first saw this mode I assumed that it was a software controlled version of SCREEN 1, but nothing could be further from the truth.

The display is in multicolor mode when bit 1 of register 0 and bit 4 of register 1 are reset and bit 3 of register 1 is set. In this mode the screen is divided into a 64 x 48 grid of colours, each the equivalent of a 4 x 4 pixel block in graphics II. In this mode, only the Pattern Name Table (0800H) and Pattern Generator Tables (0000H) are used, the Colour Table is not required.

The relationship between the two tables is a bit more complex than the previous mode. Each entry in the NTB describes a pattern block for that position. Each position defines a 2 x 2 colour block producing an 8 x 8 SCREEN 1 pixel area on the screen. Diagram 1 illustrates this relationship. There are 32 columns by 24 rows. This produces a total of 768 positions, the same number as in graphics II.




As might be expected, a byte in the NTB indicates which block of 8 bytes on the PGB is used, however it is not a straight one-one relationship as in graphics II. Only 2 bytes of each 8 byte pattern is used for each position. Refer to diagram 2.



SPECTRAVIDEO™

AUSTRALASIAN USERS GROUP

Since there are only 16 different colours, only 4 bits or half a byte is needed for each pixel. Thus the first byte is used for pixels A & B, the second byte for pixel C & D. The 4 most significant bits indicate A & C, the least indicate B & D. Which 2 bytes of the pattern block are used? This is dependent upon the row the character position is in. If the row is 0, 4, 8, 12, 16 or 20 the first 2 bytes are used, if the row is 1, 5, 9, 13, 17 or 21 the second 2 bytes are used, if the row is 2, 6, 10, 14, 18 or 22 the third 2 bytes are used, if the row is 3, 7, 11, 15, 19 or 23 the last 2 bytes are used. Diagram 3 illustrates this relationship.



This sounds more complicated than it is. The net effect is that 4 character positions, or 16 pixels, can be defined using 8 bytes of the PGB. To simplify matters listing 1 gives a visual example of the various relationships.

It would first appear that SCREEN 2 is in all ways inferior to SCREEN 1, since only a lower definition can be obtained. This is balanced by two advantages; firstly SCREEN 2 only takes up 4K of VRAM, enabling up to four pages being stored at any time. By juggling the registers you could produce a short animation; secondly GET & PUT work properly on this mode. Listing 2 demonstrates a slow scroll on SCREEN 2 using the SPRITE bug listed in an earlier article.

SPRITES


Sprites can be used in all modes except text mode. In this mode

SPECTRAVIDEO™

AUSTRALASIAN USERS GROUP

they are blanked out by the VDP. The appearance and location of sprites is dependent upon two tables: the Sprite Attribute Table (SAB) and the Sprite Pattern Generator Table (SPGB). The size and magnification is also dependent upon bits one and zero of register 1 respectfully.

The SAB starts in either SCREEN 1 or 2 at 1B00H. As there are 32 possible sprites, the SAB holds an entry for each sprite; each entry is four bytes long producing a table of 128 bytes in length. Diagram 4 shows how each entry is divided.



The first two bytes define the X-Y locations of the sprite, with the origin being the top-left corner of the screen. The first (vertical) byte is partially signed so that values of between -31 to 0 (E1 to 00 in HEX) allow a sprite to "bleed in" from the top of the screen, showing only part of the sprite. Values about 191 will do this at the bottom of the screen. A value of 208 (D0 HEX) in this byte will blank out the sprite but it will also truncate sprite processing so that sprites after that one will not be processed.


The second (horizontal) byte is not signed like the first and defines 256 positions horizontally. How then does the sprite "bleed in" from the right? The most significant bit of the fourth byte is called the Early Clock (EC) bit. When this is set to 0, it does nothing; when set to 1, the horizontal position of the sprite is shifted to the left by 32 pixels.

The third byte points to the pattern used in the SPGB. This table is explained below. The fourth byte uses the 4 least significant bits to define the colour of the sprite and the most significant bit is the early clock bit as mentioned above. The remaining bits are unused.

The SPGB is normally set to 3800H by basic. There are 256 pattern definitions in this table, each is 8 bytes long producing a table of 2048 bytes. The third byte in the SAB is used to point to a pattern position in this table and can be any of these patterns. Diagram 5 illustrates the relationships between the two tables.

SPECTRAVIDEO™

AUSTRALASIAN USERS GROUP



Bit one in register 1 (size) affects how the patterns are displayed. If set to 0, only 8 bytes are used for an 8 x 8 pattern; if set to 1 32 bytes are used for a 16 x 16 pattern, and this means that the next three patterns in the table are also used.

Bit zero of register 1 is used for magnification. When set to 0, each bit in a pattern is displayed as a 1 x 1 pixel block; when set to 1 each bit is displayed as a 2 x 2 pixel block. The important thing to remember is that size and magnification apply to ALL sprites displayed. You cannot mix sizes or magnifications at the same time. The way around this limitation is to set the size to 16 x 16 and then enlarge a pattern on a bit basis. Listing 3 illustrates manipulation of the sprite tables to produce desired effects.

Next month I will deal with the unused mode, Graphics Mode I, and explain how it differs from the implemented modes; also with GET & PUT and numeric bases.

LISTING 1

by : L.A. DUNNING

```
1 REM Listing 1, Part 6
2 REM Demonstrates Relationships in SCREEN2
3 CLEAR2000:DEFINT A-Z:SCREEN2:NT=&H800:PG=0
4 REM Set up colours in first two patten definitions
5 FORA=0TO15:VPOKEPG+A,A+A*16:NEXT
6 REM Poke in horizontal line by using patten # 0
7 FORA=2TO31:VPOKEA,A,0:NEXT
8 REM Poke in vertical line by using patten # 1
9 FORA=0TO24:VPOKEA+4+A*32,1:NEXT
10 IFNOT(STRIG(0))GOTO10
11 SCREEN1:LIST:REM You can list in SCREEN1, not very useful but interest
ing to know. Try changing line 11 to SCREEN2 instead of SCREEN1. Press
Stop when you see this line
12 REM
```

SPECTRAVIDEO™

AUSTRALASIAN USERS GROUP

LISTING 2

by : L.A. DUNNING

```
1 ' Listing 2, Part 6
2 'Demonstrates Scrolling in SCREEN2
3 'Initialisation
4 CLEAR2000:J=RND(-TIME):DEFSTRA:DEFINTV:SCREEN2,2:XL=40:YL=30:DIM A(18)
  ,V(XL,YL):NT=&H800:PG=0
5 'Putting pattens into strings
6 PRINT"1":LOCATE32,0:PRINT"2":LOCATE64,0:PRINT"3":FORB=16TO18:A(B)=RIGH
  T$(SPRITE$(48+B),31)+LEFT$(SPRITE$(49+B),1):NEXT
8 FORB=0TO15:A(B)=STRING$(32,B+B*16):NEXT
10 FORX=0TOXL:FORY=0TOYL:V(X,Y)=RND(2)*19:NEXT:NEXT:FORX=5TO7:V(X,4)=X+11
  :NEXT:X=0:Y=0
11 'Scroll Routines
12 FORX1=XTOX+7:FORY1=YTOY+5:Z=64+(X1-X)+8*(Y1-Y):SPRITE$(Z)=A(V(X1,Y1)):
  NEXT:NEXT
14 DR=STICK(0):ONDRGOSUB20,24,26,30,32,36,38,42:IFSTRIG(0)GOTO44
16 IFINKEY$(CHR$(27))THENX=0:Y=0:GOTO12
18 IFDRGOTO12ELSEGOTO14
20 Y=Y-1:IFY<0THENY=0
22 RETURN
24 GOSUB20:GOSUB26:RETURN
26 X=X+1:IFX>XL-8THENX=XL-8
28 RETURN
30 GOSUB26:GOSUB32:RETURN
32 Y=Y+1:IFY>YL-6THENY=YL-6
34 RETURN
36 GOSUB32:GOSUB38:RETURN
38 X=X-1:IFX<0THENX=0
40 RETURN
42 GOSUB38:GOSUB20:RETURN
44 REM Last line
```

LISTING 3

by : L.A. DUNNING

```
1 ' Listing 3, Part 6
2 'Demonstrates Scrolling in SCREEN2
3 'Initialisation
4 CLEAR2000:SCREEN1,0:DIMS$(4):RESTORE:SA=&H1B00:SP=&H3800
5 FORA=0TO4:READA$:FORX=1TOLEN(A$)STEP2:S$(A)=S$(A)+CHR$(VAL("&H"+MID$(A
  $,X,2))):NEXT:NEXT
6 FORA=0TO4:SPRITE$(A)=S$(A):NEXT
7 PUTSPRITE0,(-4,8),15,2:PUTSPRITE1,(60,-4),1,1:FORA=2TO4:PUTSPRITEA,(32
  *A,32*A),1,A:NEXT
8 KEY ON:ONKEYGOSUB100,200
9 LINE(0,0)-(255,50),6,BF:LOCATE0,0:PRINT"      Sprite      Y      X      Patt
  en Tag":FORA=0TO4:SX=SA+A*4:PRINTUSING"      ##      ###      ###      ###
      ###";A,VPEEK(SX),VPEEK(SX+1),VPEEK(SX+2),VPEEK(SX+3):NEXT
10 PRINT"Press F1 or F2 to change size & Mag"
11 IFNOT(STRIG(0))GOTO11ELSEGOTO40000
100 REM Change SIZE
101 GOSUB700:IFPVAND2THENPV=PVAND253ELSEPV=PVOR2
102 GOTO202
200 REM Change MAG
201 GOSUB700:IFPVAND1THENPV=PVAND254ELSEPV=PVOR1
202 GOSUB800:RETURN
700 PV=PEEK(&HFA07):RETURN
800 POKE&HFA07,PV:OUT129,PV:OUT129,129:RETURN
2000 DATA 708898A8C88897000,2068202020207000,706306304080F800,F8061030088870
  00,55AA55AA55AA55AA
40000 REM Last Line
```

SPECTRAVIDEO™

AUSTRALASIAN USERS GROUP

COPY PATCH

Last month we showed you how to decrease your disk switch off time. But I mentioned that you could not use COPY (ICOPY works O.K.), well here is a patch for copy that will fix the problem.

A>DDT COPY.COM

-S100

100	XX	C3	
101	XX	50	
102	XX	05	
103	XX	.	

-S550

550	XX	3E	
551	XX	07	
552	XX	32	
553	XX	70	*
554	XX	EF	*
555	XX	31	
556	XX	63	
557	XX	05	
558	XX	C3	
559	XX	03	
55A	XX	01	
55B	XX	.	

-^C

A>SAVE 5 COPY.COM

* NOTE: THIS PATCH IS FOR CP/M 2.2.

If you have CP/M 2.22 alter 70 EF to 50 F0.

This patch is for people who know their way around CP/M. If you have any problems, contact the LIBRARY and we will send you a PATCHED COPY program.

SPECTRAVIDEO™

AUSTRALASIAN USERS GROUP

PROGRAM OF THE MONTH (CYCLE)

by : D.F.C. NAPPER

```
10 ' CYCLE PROGRAM
20 ' WRITTEN BY
30 ' D.F.C.NAPPER
40 ' FOR S.A.U.G.
50 ' ON THE 7/8/84
60 ' Use the space bar or trigger
70 ' to accelerate or decelerate
80 ' cycle. Try to jump over as
90 ' many barrels as you can!
100 CLEAR2000
110 COLOR15,1,1
120 SCREEN1,2
130 LOCATE90,90:PRINT"BARREL JUMP" :PLAY"o4 18 g116gd18g.116 gd18gb14o5d
    ", "o418b116 bo5co418b.116bo5c "
140 DIM C$(15):DIMD$(15):DIME$(15)
150 FOR T=1 TO 13
160 FOR I=1 TO 16
170 READ A$,B$
180 C$(T)=C$(T)+CHR$(VAL(A$))
190 D$(T)=D$(T)+CHR$(VAL(B$))
200 NEXT I
210 SPRITE$(T)=C$(T)+D$(T)
220 NEXTT
230 DATA 0,0,0,0,0,0,0,0,0,0,0,0,3,0,3,0,0,0,3,16,27,248,15,136,25,230,47,
    235,53,137,24,6
240 DATA 0,0,0,0,0,0,0,0,0,0,0,0,3,0,3,0,0,0,3,16,27,248,15,136,25,230,55,
    237,45,235,24,6
250 DATA 0,0,0,0,0,0,0,0,0,0,0,0,3,0,3,0,0,96,3,144,19,102,15,235,25,237,4
    7,198,53,0,24,0
260 DATA 0,0,0,0,0,0,0,0,0,0,0,0,3,0,3,0,0,96,3,144,19,102,15,237,25,235,5
    5,198,45,0,24,0
270 DATA 0,0,0,0,0,0,0,0,0,0,0,0,64,3,78,3,72,0,248,1,240,17,18,15,180,25,24
    4,47,120,53,124,24,36
280 DATA 39,51,85,67,118,88,85,51,0,0,0,64,3,78,3,72,0,248,1,240,17,18,15,
    180,25,244,47,120,53,124,24,36
290 DATA 0,192,0,192,3,240,3,240,0,192,0,192,0,0,1,192,1,192,0,152,57,196,
    73,168,175, 224,159,253,127,252,0,0
300 DATA 0,0,168,0,212,0,168,0,212,0,128,0,128,0,128,0,128,0,252,0,199,224
    ,172,255,149,84,170,42,197,81,196,145
310 DATA 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,248,0,143,192,85,
    252,34,255
320 DATA 0,0,48,0,0,73,56,0,79,254,136,2,136,226,248,226,235,250,251,250,2
    51,251,192,227,176,237,104,26,91,214,48,12
330 DATA 131,0,52,0,179,4,56,0,79,254,136,2,136,226,248,226,235,250,251,25
    0,251,251,192,227,176,237,88,22,107,218,48,12
340 DATA 0,0,0,12,0,0,0,28,127,226,64,17,71,17,71,31,95,215,95,223,95,223,
    199,3,183,13,88,22,107,218,48,12
350 DATA 0,193,0,44,0,205,0,28,127,226,64,17,71,17,71,31,95,215,95,223,95,
    223,199,3,193,13,104,26,91,214,48,12
360 PL=4
370 F=60
```

SPECTRAVIDEO™

AUSTRALASIAN USERS GROUP

```
380 X=20:COLOR15,1,1:LINE(20,10)-(230,157),14,BF:LINE(20,57)-(240,107),1,
  B:LINE(35,156)-(50,150),13:LINE(50,150)-(50,156),13:LINE(30,156)-(50,1
  56),13:PAINT(46,154),13:CIRCLE(P,152),4,9,,1.3:PAINT(P,152),9
390 PLAY"t25516o4cfao5cr6o4ao512cr"
400 PUT SPRITE 8,(P+20,140),13:PUT SPRITE 9,(P+36,140),13:LOCATE20,2:COLO
  R15:PRINT"BARRELS JUMPED:-";BA:LOCATE 148,2:PRINT"RIDERS LEFT:";PL-1:L
  INE(0,160)-(255,192),1,BF
410 PUT SPRITE 1,(X,40),1,1:COLOR1:LOCATE 30,20:PRINT"█ SET":LOCATE30,28:
  PRINT"|":FOR T=1 TO400:NEXT:LOCATE30,20:COLOR14:PRINT"██████████":LOCAT
  E30,28:PRINT"█":LOCATE 30,28:COLOR 1:PRINT"_█ GO!":FOR T=1 TO100:NE
  XT:LINE(30,20)-(80,45),14,BF
420 SOUND7,56
430 SOUND6,12:SOUND8,12:SOUND9,12:SOUND10,12
440 SOUND1,1:SOUND3,2:SOUND5,5
450 Y=40
460 IF A>12 THENA=12
470 IF Y>145 THENY=40
480 IF X>200THENY=Y+50:X=10
490 S=INT(220/(A+4)):IF S>100THENS=100
500 Z=S*2:SOUND0,Z:SOUND4,Z:SOUND2,Z
510 IF POINT(X+16,Y+16)=13THENSOUND0,10:SOUND4,10:SOUND2,10:GOSUB570
520 J=STRIG(0)+STRIG(1):IF J=-1 THEN A=A+.2:IF A>11 THEN GOTO 560ELSE:X=X
  +1*A:PUT SPRITE1,(X+1,Y),1,3:FOR T=1 TO10:NEXT:PUT SPRITE1,(X+1,Y),1
  ,4:FOR T=1 TO10:NEXT:GOTO 460
530 A=A-.4:IF A<1 THENA=1ELSE:X=X+1*A:PUT SPRITE 1,(X,Y),1,1:FOR T=1 TO1
  0:NEXT:PUT SPRITE 1,(X+1,Y),1,2:FOR T=1 TO10:NEXT
540 IF X>200THENY=Y+50:X=10
550 GOTO 460
560 X=X+1*A:PUT SPRITE 1,(X,Y),1,1:FOR T=1 TO10:NEXT:PUT SPRITE 1,(X+2,Y
  ),1,2:FOR T=1 TO10:NEXT:GOTO 460
570 X1=X
580 IF A<6THENA=6
590 X2=A*X1/1.3
600 FOR X=X TOX2STEP.3*A
610 PUT SPRITE1,(X,Y),1,4
620 IF X=>9*X2/16THENY=Y+.09*AELSEY=Y-.09*A
630 IF Y>150 THEN Y=150
640 IF POINT(X,Y+16)=9 THEN GOSUB700
650 IF X>P+50AND Y<146 THEN GOSUB700
660 PUT SPRITE1,(X,Y),1,3
670 NEXTX:P=P+10:BA=BA+1:CIRCLE(P,152),4,9,,1.3:PAINT(P,150),9:PUT SPR
  ITE 8,(P+20,140),13:PUT SPRITE9,(P+36,140),13
680 IF BA>11 THEN LOCATE80,40:COLOR 1:PRINT"CONGRATULATIONS !
  YOU DID IT ,12 barrels wow!":SOUND7,62:PLAY"o218dafeco612ac
  ge":GOTO780
690 LOCATE119,2:COLOR0:PRINT"███":LOCATE 115,2:COLOR15:PRINTBA:RETURN
700 SOUND6,15:SOUND7,7:SOUND8,16:SOUND9,16:SOUND10,16:SOUND12,16:SOUND13,0
710 PUT SPRITE1,(X,Y),1,6:FOR R=1 TO1000:NEXT
720 PUT SPRITE1,(X,Y),1,5:FOR R=1 TO100:NEXT:PUT SPRITE1,(X+5,Y),1,7:PL=P
  L-1:
730 SOUND0,255:SOUND7,62:SOUND8,15:FOR X1=240TO X+10STEP-2
740 PUT SPRITE 2,(X1,140),8,10:FOR T=1 TO20:SOUND0,T:NEXT:PUT SP
  RITE 2,(X1-1,140),8,11:SOUND0,80:FOR T=1 TO40STEP-1:NEXT:NEXTX1:PUT SP
  RITE2,(X,209)
750 FOR X2=X TO255STEP2:PUT SPRITE1,(X2,140),8,12:FOR T=1 TO20:NEXT:SOUND0,
  X2:PUT SPRITE1,(X2,140),8,13:FOR T=1 TO20:NEXT:NEXTX2
760 IF PL<0 THENCOLOR15:PLAY"t110o3:4dr64di64r64d14r64dr64f160r64e16r64er
  64dr64dr64c+r64dr2o7181ddefo6bago7c":LOCATE20,164:PRINT"HOSPITAL NOTIC
  E:- Sorry,your rider was dead on arrival":GOTO
  780
```

SPECTRAVIDEO™

AUSTRALASIAN USERS GROUP

```
770 COLOR15:PLAY"t110o313dfd120o7t25516o5cfao6cr6l6o5ao6l1c":LOCATE20,160:
PRINT"HOSPITAL NOTICE:- YOUR RIDER HAS SUR
VIVED, GET READY":FOR T=1 TO800:NEXT:LOCATE217,2:COLOR0:PRINT"■■■":X=2
0:A=0:FORT=1TO200:NEXT:GOTO400
780 LOCATE 40,183:COLOR15:PRINT"DO YOU WISH TO TRY AGAIN (Y/N) ":FOR T=1 T
05000:NEXT
800 A$=INKEY$:IF A$="y" OR A$="Y" THENCLS:A=0: GOTO360
810 IF A$="N"ORA$="n"THENEND
820 GOTO 800
```

GAMES GALORE
SHOP 5
358 MAIN ROAD
GLENORCHY 7010
PH: 720424

SV-318 / CASSETTE / 4 x PROGRAMS = \$299
SV-328 / " / " = \$599
DISK DRIVE = \$499
EXPANDER = \$199

E.T.C.

S-V RETAILERS

To assist all members the Newsletter wishes to print a list of ALL the retail outlets throughout Australia & New Zealand. We require your help in finding all the Names and Addresses needed. So please contact your local dealer and get him to pass the relevent information on to us. We will only print the information if we receive it from the Retail outlet DIRECT.

SPECTRAVIDEO™

AUSTRALASIAN USERS GROUP

GRAPHICS TABLET

By D. Cooper

First impressions of the SV105 Graphics Tablet, as tested by David Cooper of Coastal Computers.

A demonstration unit loaned by Rose Music came without the documentation so we had to operate on a 'see and do' basis. The unit measures 340 x 280 mm with a working surface of 185 x 155 mm. It connects to Computer joystick port 1 and the operating system software comes on cassette. BLOAD"tablet",R produces a menu on screen at completion of load.

This menu allows easy selection of any one of twelve operations and anyone of fifteen colors. The options are shown pictorially and with a single word description as listed below :-

DRAW	GROUND	ERASE	TEXT	=/=
BOX	CIRCLE	PAINT	RULER	(COLORS)
PRINT	LINE	CSAVE	CLOAD	=/=

A 'PENCIL' is moved around this menu by using the cursor keys. The 'SELECT' key enables the chosen function which flashes on screen to show it has been selected. The selected color also flashes on screen.

You use the 'ESCAPE' key to toggle between the menu and the working screen.

Brief descriptions of each operation follow :-

1. DRAW Freehand drawing in the selected color. Wherever the tablet is touched a line is drawn on the screen. Could be great for tracing over a picture.
2. GROUND Sets background to selected color.
3. ERASE Clears small blocks of the screen wherever tablet is touched.
4. TEXT Allows typing of text on the screen with all normal full-screen editing features.
5. BOX Draws a box defined by two opposite corners.
6. CIRCLE Draws a circle defined by the centre and any point on the circumference.
7. PAINT Select a point within the area you wish to paint and press the SELECT key to fill the area with the currently selected color.
8. RULER A 'pencil' on the screen is moved with the cursor keys and is toggled on/off with the SELECT key. When 'on' it draws a line when moved and will generate lines horizontally, vertically, and at 45 degree angles.
9. PRINT Dump screen contents to suitable printer.
10. LINE Draws a straight line between two points marked on screen.
11. CSAVE Saves screen contents to cassette with chosen filename.
12. CLOAD Loads a screen from cassette. There were two demo programs on our tape. One of the Space Shuttle, the other showed the map of Pacific and Indian Oceans centered on Australia.

It would appear to be well made, good quality, and with a little imagination could be used to drive the SV318/328 s to the limit of their graphics ability.

SPECTRAVIDEO™

AUSTRALASIAN USERS GROUP

LIBRARY NOTES

By J. Collins.


Quite a busy month for the library.....many members have made use of the services and some new programs have been received for possible inclusion and sale. Those titles will appear in a forthcoming newsletter.

One small item was left out of the last newsletter because of an oversight. The program listed as Mysterious Manor is over 27 KBytes long and as such is not suitable for use with an unexpanded SV318 or for use with an SV328 using disk basic, unless extra memory has been installed. It can be used with a standard 328 under cassette basic. My apologies for the omission.

ITEM FOR SALE

I have a SIEKO GP80 80 column printer for sale. It is in as new condition with a brand new ribbon and full documentation. This is an excellent small dot matrix printer which operates at 30 CPS and produces very readable printout. It does not have all of the bells and whistles of the many EPSON clones but it is ideal for all general purpose printing and is especially suitable for program listing and development work which is why I first bought it. Now that I have become involved in the newsletter and so on I need a more professional printer and will part with my GP80 to a good home for a very reasonable \$215.00. (It cost me \$499.00 when I bought it) Enquiries per mail to the address for the library. You buy the printer and I will pay Courier charges.

Also for sale (or possible swap) is a brand new copy of 'SPECTRON'. I am informed by the present owner of this program that he will sell for a very reasonable figure or will swap for a program he does not have. Interested members should write direct to Mr Michael Bourrillon of 6/142 Hotham Street, East St Kilda in Victoria. Post Code 3183.



SPI01 PARALLEL PRINTER INTERFACE

- SUITS SPECTRAVIDEO SV318 - SV328



The SP101 simply plugs into the back of the computer and accepts a standard Centronics printer cable which can be obtained locally or we can supply one at a very competitive price.

Now for the good news. As an introductory offer, for September and October, we can offer members of your user group 10% off the normal retail price. (Unfortunately Australia Post will still want the normal amount for postage).

ORDER FORM

FORREST DATA SERVICES P.O. Box 71 Palmyra WA 6157
(09) 339 5087

Please send me the following

Qty	Product	Amount
.....	SPI01 PARALLEL PRINTER INTERFACE @ \$70
.....	CENTRONICS PRINTER CABLE @ \$35
	Plus postage & packing (\$2 per item)
	Plus \$1 Airmail per item (if required)
	(eg 1 interface & 1 cable by airmail \$111)
	I ENCLOSE CHEQUE/MONEY ORDER FOR \$

NAME Please
 ADDRESS Print
 P/code.....

Allow 14 days delivery

SPECIFICATIONS

- NO MODIFICATIONS TO THE COMPUTER
- NO SEPARATE POWER SUPPLY REQUIRED
- ATTRACTIVE RIGID WHITE ACRYLIC CASE
- SUITS SV318 AND SV328 COMPUTERS
- ACCEPTS STANDARD CENTRONICS PRINTER CABLE
- DIMENSIONS 105W x 27H x 85D
- WEIGHT 100 GRAM
- CENTRONICS PRINTER CABLE AVAILABLE (1 metre)

FITTING INSTRUCTIONS -

SIMPLY PLUG PRINTER CABLE INTO THE PRINTER INTERFACE WHICH THEN PLUGS INTO THE BACK OF THE COMPUTER.

P.S. Please advise members to quote their membership number when placing an order.